NAG Library Routine Document

F08ZPF (ZGGGLM)

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

F08ZPF (ZGGGLM) solves a complex general Gauss–Markov linear (least squares) model problem.

2 Specification

```
SUBROUTINE F08ZPF (M, N, P, A, LDA, B, LDB, D, X, Y, WORK, LWORK, INFO)

INTEGER                 M, N, P, LDA, LDB, LWORK, INFO
COMPLEX (KIND=nag_wp) A(LDA,*), B(LDB,*), D(M), X(N), Y(P),           &
                        WORK(max(1,LWORK))
```

The routine may be called by its LAPACK name ***zggglm***.

3 Description

F08ZPF (ZGGGLM) solves the complex general Gauss–Markov linear model (GLM) problem

$$\underset{x}{\text{minimize}} \|y\|_2 \qquad \text{subject to} \qquad d = Ax + By$$

where A is an m by n matrix, B is an m by p matrix and d is an m element vector. It is assumed that $n \leq m \leq n + p$, $\text{rank}(A) = n$ and $\text{rank}(E) = m$, where $E = \begin{pmatrix} A & B \end{pmatrix}$. Under these assumptions, the problem has a unique solution x and a minimal 2-norm solution y, which is obtained using a generalized QR factorization of the matrices A and B.

In particular, if the matrix B is square and nonsingular, then the GLM problem is equivalent to the weighted linear least squares problem

$$\underset{x}{\text{minimize}} \left\| B^{-1}(d - Ax) \right\|_2.$$

4 References

Anderson E, Bai Z, Bischof C, Blackford S, Demmel J, Dongarra J J, Du Croz J J, Greenbaum A, Hammarling S, McKenney A and Sorensen D (1999) *LAPACK Users' Guide* (3rd Edition) SIAM, Philadelphia

Anderson E, Bai Z and Dongarra J (1992) Generalized *QR* factorization and its applications *Linear Algebra Appl. (Volume 162–164)* 243–271

5 Parameters

1: M – INTEGER *Input*

On entry: m, the number of rows of the matrices A and B.

Constraint: $\text{M} \geq 0$.

2: N – INTEGER *Input*

On entry: n, the number of columns of the matrix A.

Constraint: $0 \leq \text{N} \leq \text{M}$.

3: P – INTEGER *Input*

On entry: p, the number of columns of the matrix B.

Constraint: $P \geq M - N$.

4: A(LDA,$*$) – COMPLEX (KIND=nag_wp) array *Input/Output*

Note: the second dimension of the array A must be at least $\max(1, N)$.

On entry: the m by n matrix A.

On exit: A is overwritten.

5: LDA – INTEGER *Input*

On entry: the first dimension of the array A as declared in the (sub)program from which F08ZPF (ZGGGLM) is called.

Constraint: $LDA \geq \max(1, M)$.

6: B(LDB,$*$) – COMPLEX (KIND=nag_wp) array *Input/Output*

Note: the second dimension of the array B must be at least $\max(1, P)$.

On entry: the m by p matrix B.

On exit: B is overwritten.

7: LDB – INTEGER *Input*

On entry: the first dimension of the array B as declared in the (sub)program from which F08ZPF (ZGGGLM) is called.

Constraint: $LDB \geq \max(1, M)$.

8: D(M) – COMPLEX (KIND=nag_wp) array *Input/Output*

On entry: the left-hand side vector d of the GLM equation.

On exit: D is overwritten.

9: X(N) – COMPLEX (KIND=nag_wp) array *Output*

On exit: the solution vector x of the GLM problem.

10: Y(P) – COMPLEX (KIND=nag_wp) array *Output*

On exit: the solution vector y of the GLM problem.

11: WORK(max(1, LWORK)) – COMPLEX (KIND=nag_wp) array *Workspace*

On exit: if $INFO = 0$, the real part of WORK(1) contains the minimum value of LWORK required for optimal performance.

12: LWORK – INTEGER *Input*

On entry: the dimension of the array WORK as declared in the (sub)program from which F08ZPF (ZGGGLM) is called.

If $LWORK = -1$, a workspace query is assumed; the routine only calculates the optimal size of the WORK array, returns this value as the first entry of the WORK array, and no error message related to LWORK is issued.

Suggested value: for optimal performance, $LWORK \geq N + \min(M, P) + \max(M, P) \times nb$, where nb is the optimal ***block size***.

Constraint: $LWORK \geq \max(1, M + N + P)$ or $LWORK = -1$.

13: INFO – INTEGER *Output*

On exit: INFO $= 0$ unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

Errors or warnings detected by the routine:

INFO < 0

If INFO $= -i$, argument i had an illegal value. An explanatory message is output, and execution of the program is terminated.

INFO $= 1$

The upper triangular factor R associated with A in the generalized RQ factorization of the pair (A, B) is singular, so that $\text{rank}(A) < m$; the least squares solution could not be computed.

INFO $= 2$

The bottom $(N - M)$ by $(N - M)$ part of the upper trapezoidal factor T associated with B in the generalized QR factorization of the pair (A, B) is singular, so that $\text{rank}\begin{pmatrix} A & B \end{pmatrix} < N$; the least squares solutions could not be computed.

7 Accuracy

For an error analysis, see Anderson *et al.* (1992). See also Section 4.6 of Anderson *et al.* (1999).

8 Further Comments

When $p = m \geq n$, the total number of real floating point operations is approximately $\frac{8}{3}\left(2m^3 - n^3\right) + 16nm^2$; when $p = m = n$, the total number of real floating point operations is approximately $\frac{56}{3}m^3$.

9 Example

This example solves the weighted least squares problem

$$\underset{x}{\text{minimize}}\left\|B^{-1}(d - Ax)\right\|_2,$$

where

$$B = \begin{pmatrix} 0.5 - 1.0i & & & \\ & 1.0 - 2.0i & & \\ & & 2.0 - 3.0i & \\ & & & 5.0 - 4.0i \end{pmatrix},$$

$$d = \begin{pmatrix} 6.00 - 0.40i \\ -5.27 + 0.90i \\ 2.72 - 2.13i \\ -1.30 - 2.80i \end{pmatrix}$$

and

$$A = \begin{pmatrix} 0.96 - 0.81i & -0.03 + 0.96i & -0.91 + 2.06i \\ -0.98 + 1.98i & -1.20 + 0.19i & -0.66 + 0.42i \\ 0.62 - 0.46i & 1.01 + 0.02i & 0.63 - 0.17i \\ 1.08 - 0.28i & 0.20 - 0.12i & -0.07 + 1.23i \end{pmatrix}.$$

Note that the block size (NB) of 64 assumed in this example is not realistic for such a small problem, but should be suitable for large problems.

9.1 Program Text

```
    Program f08zpfe

!     F08ZPF Example Program Text

!     Mark 24 Release. NAG Copyright 2012.

!     .. Use Statements ..
      Use nag_library, Only: dznrm2, nag_wp, zggglm
!     .. Implicit None Statement ..
      Implicit None
!     .. Parameters ..
      Integer, Parameter               :: nb = 64, nin = 5, nout = 6
!     .. Local Scalars ..
      Real (Kind=nag_wp)               :: rnorm
      Integer                          :: i, info, lda, ldb, lwork, m, n, p
!     .. Local Arrays ..
      Complex (Kind=nag_wp), Allocatable :: a(:,:), b(:,:), d(:), work(:),     &
                                          x(:), y(:)
!     .. Executable Statements ..
      Write (nout,*) 'F08ZPF Example Program Results'
      Write (nout,*)
!     Skip heading in data file
      Read (nin,*)
      Read (nin,*) m, n, p
      lda = m
      ldb = m
      lwork = n + m + nb*(m+p)
      Allocate (a(lda,n),b(ldb,p),d(m),work(lwork),x(n),y(p))

!     Read A, B and D from data file

      Read (nin,*)(a(i,1:n),i=1,m)
      Read (nin,*)(b(i,1:p),i=1,m)
      Read (nin,*) d(1:m)

!     Solve the weighted least-squares problem

!     minimize ||inv(B)*(d - A*x)|| (in the 2-norm)

!     The NAG name equivalent of zggglm is f08zpf
      Call zggglm(m,n,p,a,lda,b,ldb,d,x,y,work,lwork,info)

!     Print least-squares solution

      Write (nout,*) 'Weighted least-squares solution'
      Write (nout,99999) x(1:n)

!     Print residual vector y = inv(B)*(d - A*x)

      Write (nout,*)
      Write (nout,*) 'Residual vector'
      Write (nout,99998) y(1:p)

!     Compute and print the square root of the residual sum of squares
!     The NAG name equivalent of dznrm2 is f06jjf
      rnorm = dznrm2(p,y,1)

      Write (nout,*)
      Write (nout,*) 'Square root of the residual sum of squares'
      Write (nout,99997) rnorm

99999 Format (3(' (',F9.4,',',F9.4,')':))
99998 Format (3(' (',1P,E9.2,',',1P,E9.2,')':))
99997 Format (1X,1P,E10.2)
    End Program f08zpfe
```

9.2 Program Data

```
F08ZPF Example Program Data

  4             3             4                         :Values of M, N and P

( 0.96,-0.81) (-0.03, 0.96) (-0.91, 2.06)
(-0.98, 1.98) (-1.20, 0.19) (-0.66, 0.42)
( 0.62,-0.46) ( 1.01, 0.02) ( 0.63,-0.17)
( 1.08,-0.28) ( 0.20,-0.12) (-0.07, 1.23)               :End of matrix A

( 0.50,-1.00) ( 0.00, 0.00) ( 0.00, 0.00) ( 0.00, 0.00)
( 0.00, 0.00) ( 1.00,-2.00) ( 0.00, 0.00) ( 0.00, 0.00)
( 0.00, 0.00) ( 0.00, 0.00) ( 2.00,-3.00) ( 0.00, 0.00)
( 0.00, 0.00) ( 0.00, 0.00) ( 0.00, 0.00) ( 5.00,-4.00) :End of matrix B

( 6.00,-0.40)
(-5.27, 0.90)
( 2.72,-2.13)
(-1.30,-2.80)                                           :End of vector d
```

9.3 Program Results

```
 F08ZPF Example Program Results

 Weighted least-squares solution
 (  -0.9846,   1.9950) (   3.9929,  -4.9748) (  -3.0026,   0.9994)

 Residual vector
 ( 1.26E-04,-4.66E-04) ( 1.11E-03,-8.61E-04) ( 3.84E-03,-1.82E-03)
 ( 2.03E-03, 3.02E-03)

 Square root of the residual sum of squares
   5.79E-03
```